// File: Port3BQLSharesFactory.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.8.0;

import "Ownable.sol";
import "Port3BQLSharesProxy.sol";

contract Port3BQLSharesFactory is Ownable {

    address public tokenImplementation;

    event ImplChanged(address newImplementation);
    event ProxyCreated(address indexed proxy, address deployer);

    constructor(address _bqlSharesImpl, address _owner) Ownable(_owner) {
        tokenImplementation = _bqlSharesImpl; 
        emit ImplChanged(_bqlSharesImpl);
    }

    function setImpl(address newImpl) external onlyOwner {
        tokenImplementation = newImpl;
        emit ImplChanged(newImpl);
    }

    function deployBQLShares(
        string memory _name,
        string memory _symbol, 
        string memory _uri,
        address _sharesSubject,
        address _protocolFeeDestination,
        uint256 _curveBase 
    ) external onlyOwner {
        // owner, _name, _symbol, _uri, _sharesSubject, _protocolFeeDestination, _curveBase
        bytes memory params = abi.encode(msg.sender, _name, _symbol, _uri, _sharesSubject, _protocolFeeDestination, _curveBase);
        _deploy(params);
    }

    function _generateInitData(bytes memory data) private pure returns (bytes memory res) {
        bytes4 selector;
        selector = 0xea2b3316; // abi selector for initializer 
        res = abi.encodePacked(selector, data);
    }

    function _deploy(
        bytes memory params
    ) private returns (address) {
        bytes memory initData = _generateInitData(params);
        bytes memory callData = abi.encode(tokenImplementation, initData);
        bytes memory bytecode = abi.encodePacked(type(Port3BQLSharesProxy).creationCode, callData);
        bytes32 salt = keccak256(abi.encodePacked(msg.sender, params));
        address res;
        assembly {
            res := create2(0, add(bytecode, 0x20), mload(bytecode), salt)
        }
        require(res != address(0), "Shares failed to deploy");
        emit ProxyCreated(res, msg.sender);
        return res;
    }

}


// File: Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (access/Ownable.sol)

pragma solidity ^0.8.20;

import {Context} from "Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * The initial owner is set to the address provided by the deployer. This can
 * later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    /**
     * @dev The caller account is not authorized to perform an operation.
     */
    error OwnableUnauthorizedAccount(address account);

    /**
     * @dev The owner is not a valid owner account. (eg. `address(0)`)
     */
    error OwnableInvalidOwner(address owner);

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the address provided by the deployer as the initial owner.
     */
    constructor(address initialOwner) {
        if (initialOwner == address(0)) {
            revert OwnableInvalidOwner(address(0));
        }
        _transferOwnership(initialOwner);
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        if (owner() != _msgSender()) {
            revert OwnableUnauthorizedAccount(_msgSender());
        }
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        if (newOwner == address(0)) {
            revert OwnableInvalidOwner(address(0));
        }
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (utils/Context.sol)

pragma solidity ^0.8.20;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: Port3BQLSharesProxy.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.8.0;

import "Proxy.sol";

contract Port3BQLSharesProxy is Proxy {
    address private immutable _impl;

    constructor(address _logic, bytes memory _data) {
        (bool success, ) = _logic.delegatecall(_data);
        require(success, "Token failed to initialize");
        _impl = _logic;
    }

    function _implementation() internal view override returns (address) {
        return _impl;
    }

    function implementation() external view returns (address) {
        return _impl;
    }
}

// File: Proxy.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (proxy/Proxy.sol)

pragma solidity ^0.8.20;

/**
 * @dev This abstract contract provides a fallback function that delegates all calls to another contract using the EVM
 * instruction `delegatecall`. We refer to the second contract as the _implementation_ behind the proxy, and it has to
 * be specified by overriding the virtual {_implementation} function.
 *
 * Additionally, delegation to the implementation can be triggered manually through the {_fallback} function, or to a
 * different contract through the {_delegate} function.
 *
 * The success and return data of the delegated call will be returned back to the caller of the proxy.
 */
abstract contract Proxy {
    /**
     * @dev Delegates the current call to `implementation`.
     *
     * This function does not return to its internal call site, it will return directly to the external caller.
     */
    function _delegate(address implementation) internal virtual {
        assembly {
            // Copy msg.data. We take full control of memory in this inline assembly
            // block because it will not return to Solidity code. We overwrite the
            // Solidity scratch pad at memory position 0.
            calldatacopy(0, 0, calldatasize())

            // Call the implementation.
            // out and outsize are 0 because we don't know the size yet.
            let result := delegatecall(gas(), implementation, 0, calldatasize(), 0, 0)

            // Copy the returned data.
            returndatacopy(0, 0, returndatasize())

            switch result
            // delegatecall returns 0 on error.
            case 0 {
                revert(0, returndatasize())
            }
            default {
                return(0, returndatasize())
            }
        }
    }

    /**
     * @dev This is a virtual function that should be overridden so it returns the address to which the fallback
     * function and {_fallback} should delegate.
     */
    function _implementation() internal view virtual returns (address);

    /**
     * @dev Delegates the current call to the address returned by `_implementation()`.
     *
     * This function does not return to its internal call site, it will return directly to the external caller.
     */
    function _fallback() internal virtual {
        _delegate(_implementation());
    }

    /**
     * @dev Fallback function that delegates calls to the address returned by `_implementation()`. Will run if no other
     * function in the contract matches the call data.
     */
    fallback() external payable virtual {
        _fallback();
    }
}